TU München, Fakultät für Informatik
Lehrstuhl III: Datenbanksysteme
Prof. Alfons Kemper, Ph.D.

Übung zur Vorlesung *Einsatz und Realisierung von Datenbanken* im SoSe25

Alice Rey, Maximilian Reif, Tobias Goetz (i3erdb@in.tum.de)
http://db.in.tum.de/teaching/ss25/impldb/

Blatt Nr. 04

Hinweise Die Datalogaufgaben können auf `https://souffle.db.in.tum.de/` getestet werden. Auf der Seite kann unter *examples* ein entsprechender Datensatz geladen werden. Die neuen IDB Regeln sollten am Ende der EDB definiert und dann im Query-Eingabefeld abgefragt werden.

Zusätzlich zu der in der Vorlesung vorgestellten Syntax hier noch eine Kurzübersicht der Vergleichsoperatoren: $X < Y, Y > X$ (kleiner, größer), $X <= Y, X >= Y$ (kleiner oder gleich, größer oder gleich), $X = Y, X! = Y$ (gleich, ungleich), $!pred(X, Y)$ (existiert nicht pred(X,Y)).

Hausaufgabe 1

Die Prüfungs-Datenbank der AMU wurde leider von einem unachtsamen Programmierer geschrieben. Es gibt ein Formular, in dem man nach seinen Klausurnoten suchen kann, allerdings wird die Benutzereingabe nicht geprüft.

Schema: Prüfung: {[Vorlesung, Note, Matrikelnummer]}

Benutzte Anfrage:

```
SELECT *
FROM Prüfung
WHERE Matrikelnummer='{MatrikelNr}' AND Vorlesung='{Benutzereingabe}'
```

- Benutzereingabe: ist die Benutzereingabe.
- MatrikelNr: wird automatisch mit deiner Matrikelnummer befüllt.

Schreibe eine Benutzereingabe, mit der du alle deine Noten auf 1.0 setzen kannst.

Lösung:

```
egal'; UPDATE Prüfung SET Note=1.0 WHERE Matrikelnummer='03670815';--
```

Hausaufgabe 2

Sie wollen der AMU helfen, ihre Datenbank sicherer zu machen, und finden bei einer kurzen Suche im Internet *Prepared Statements* und *Input Sanitization*.

1. Erklären Sie kurz beide Methoden, besonders deren Unterschiede in der Behandlung von bösartigen Eingaben.

Lösung:

Input Sanitization ersetzt oder entfernt Sonderzeichen, sodass die Zeichenkette sicher an die Datenbank übergegeben werden kann. Damit wird der Angriff unschädlich gemacht, da z.B. Anführungszeichen ersetzt werden.

Prepared Statement bereitet eine SQL-Anfrage mit Platzhaltern vor. Damit wird zwischen der Struktur der Anfrage und der Eingabe unterschieden. Die Eingabe wird nachträglich in den Platzhalter eingefügt. Damit kann die Struktur der Anfrage nicht mehr verändert werden und der Angriff schlägt fehl.

2. Beschreiben Sie Vorteile von *Prepared Statements*, die über Sicherheit hinaus reichen.

   Lösung: Das Vorbereiten der Anfrage kann der Datenbank erlauben, die Anfrage als Template vorzubereiten, in das nur noch an bestimmten Stellen Eingaben eingefügt werden müssen. Dadurch können Anfragen, die das gleiche Template benutzen, mit kürzerer Latenz ausgeführt werden.

3. Nachfolgend sehen Sie die Funktion `exec_unsafe()`, die das Formular serverseitig aufruft, um die Klausurnote abzufragen. Ersetzen Sie diese durch eine Funktion `exec_prep()`, die mittels eines *Prepared Statements* auf die Datenbank zugreift.

```
#include <pqxx/pqxx>
#include <iostream>
#include <string>
pqxx::result exec_unsafe(pqxx::connection& conn, int matrnr, std::string vorl){
  std::string q = "SELECT␣*␣FROM␣pruefung␣WHERE␣matrikelnummer=",
    q2 = "AND␣vorlesung='", q3 ="'";
  pqxx::work tx{conn, ""}; // Begin of transaction
  pqxx::result r(tx.exec(q + std::to_string(matrnr) + q2 + vorl + q3));
  tx.commit(); // Commit transaction
  return r;
}
int main(int argc, char* argv[]){
  pqxx::connection conn;
  auto r = exec_unsafe(conn, 123, "Grundzuege");
  for(auto row: r)
    std::cout << row["note"] << std::endl;
  return 0;
}
```

Lösung:

```
#include <pqxx/pqxx>
#include <iostream>
#include <string>
pqxx::result exec_prep(pqxx::connection& conn, int matrnr, std::string vorl){
  pqxx::work tx{conn, ""}; // Begin of transaction
  auto r = tx.prepared("getGrade")(matrnr)(vorl).exec();
  tx.commit(); // Commit transaction
  return r;
}
int main(int argc, char* argv[]){
  pqxx::connection conn;
  conn.prepare("getGrade",
    "SELECT␣*␣FROM␣pruefung␣WHERE␣matrikelnummer=$1␣AND␣vorlesung=$2");
  auto r = exec_prep(conn, 123, "Grundzuege");
  for(auto row: r)
    std::cout << row["note"] << std::endl;
  return 0;
}
```

Hausaufgabe 3

Bob hat ein Vorlesungsverzeichnis für die Universität programmiert und unter `http://db.in.tum.de/~schuele/sql_verzeichnis.html` online gestellt.

Um die Suche zu erleichtern, kann die Anzahl der SWS durch einen Parameter eingeschränkt werden. Finden Sie einen speziell präparierten Parameter, bei dessen Eingabe statt der Vorlesungen die Liste der Studenten ausgegeben wird. Die Datenbank folgt dem bekannten Universitätsschema.

Bob erfährt von der Sicherheitslücke und schlägt vor, die bekannten Tabellen einmalig mit zufälligen Namen umzubenennen, so seien sie nicht zu finden. Würde diese *Sicherheitsmaßnahme* helfen?

- Injection: `0 union all select name, matrnr, semester from studenten`
- Nein, da z.B. mit `0 union select tablename,1,1 from pg_tables` eine Liste der Datenbanken ausgegeben werden kann.

Hausaufgabe 4

Sie haben die User-Tabelle zweier Pizzalieferanten ausgelesen, jedoch scheinen die Passwörter uncharakteristisch kompliziert zu sein. Das von Ihnen erhaltene Resultat ist das folgende:

```
id          name        password
----------  ----------  ----------------------------------------
1           luigi       4d75e8db6a4b6205d0a95854d634c27a
2           mario       fe78ea401158dd5847c4090b8bb22477e510febf
```

- Was könnte der Grund für diese hexadezimalen, 32 bzw. 40 Stellen langen Passwörter sein?
- Können Sie trotzdem den Klartext finden?
- Wie können Sie das Passwort sicherer speichern?
- Wie können Sie für diese Art von Passwortspeicherung Bruteforce-Attacken erschweren?
- Das erste Passwort wurde MD5 gehasht, das zweite SHA-1.
- Webrecherche nach dem Hash, es gibt sog. Rainbow-Tables in denen zahlreiche MD5- und SHA-1-Hashes vorberechnet sind.
- MD5 mehrfach anwenden, besser: Einen Salt verwenden, beispielsweise das Passwort zusammen mit dem Erstellungsdatum des Accounts oder dem Accountnamen hashen.
- Eine bessere Hashfunktion verwenden (MD5 und SHA1 werden nicht mehr empfohlen), die mehr Rechenzeit benötigt. Genauso wichtig, den User zwingen, komplexere Passwörter zu benutzen, damit Wörterbuchangriffe ineffizient werden.
- Alternativ hilft eine Key-Derivation-Function (KDF), ein Passwort mittels einer Pseudozufallsfunktion zu strecken.

Hausaufgabe 5

Gegeben sei die folgende Segler-Boots-Reservierung-Datenbank:

```
.decl segler(sid: number, sname: symbol, einstufung: number, alter: number)
.decl boot(bid: number, bname: symbol, farbe: symbol)
.decl reservierung(sid: number, bid: number, datum: number)
```

Beantworten Sie die folgenden Anfragen in Datalog und testen Sie unter (`http://souffle.db.in.tum.de/`, Examples => Segler-Boots-Reservierung):

1. Geben Sie die Farben aller Boote, die von 'Lubber' reserviert wurden, aus.

   ```
   .decl lubber_farbe(farbe: symbol)
   lubber_farbe(F) :- segler(SID,"Lubber",_,_), reservierung(SID,BID,_),
                      boot(BID, _, F).
   .output lubber_farbe
   ```

2. Geben Sie alle Segler aus, die eine Einstufung von mindestens 8 oder das Boot 103 reserviert haben.

   ```
   .decl a2(sid: number, name: symbol)
   a2(SID,N) :- segler(SID,N,R,_), R>=8.
   a2(SID,N) :- segler(SID,N,_,_), reservierung(SID,103,_).
   .output a2
   ```

3. Geben Sie die Namen aller Segler aus, die mindestens zwei Boote reserviert haben.

   ```
   .decl doppelBoot(name: symbol)
   doppelBoot(S) :- segler(SID,S,_,_), reservierung(SID,BIDA,_),
                    reservierung(SID,BIDB,_), BIDA!=BIDB .
   ```

4. Geben Sie alle Segler aus, die noch nie ein rotes Boot reserviert haben.

   ```
   .decl rotReserviert(sid: number)
   rotReserviert(SID) :- segler(SID,_,_,_), reservierung(SID,BID,_),
                         boot(BID,_,"red").
   .decl nichtRot(sid: number, name: symbol)
   nichtRot(SID,S) :- segler(SID,S,_,_), !rotReserviert(SID).
   .output nichtRot
   ```

5. Geben Sie alle Segler aus, die mehr als 20 Jahre alt sind und kein rotes Boot reserviert haben.

   ```
   .decl rotReserviert(sid: number)
   rotReserviert(SID) :- segler(SID,_,_,_), reservierung(SID,BID,_),
                         boot(BID,_,"red").
   .decl nichtRotAlt(sid: number, segler: symbol, alter: number)
   nichtRotAlt(SID,S,A) :- segler(SID,S,_,A), A>20, !rotReserviert(SID).
   .output nichtRotAlt
   ```

6. Geben Sie die Ids der Segler aus, deren Einstufung besser als die eines Seglers mit Namen 'Horatio' ist.

   ```
   .decl nichtSchlecht(sid: number)
   nichtSchlecht(SID) :- segler(SID,_,R,_), segler(_,"Horatio",RH,_), R > RH.
   .output nichtSchlecht
   ```

7. Geben Sie die Ids der Segler aus, deren Einstufung besser als die aller Segler mit Namen 'Horatio' ist.

```
.decl dochSchlecht(sid: number)
dochSchlecht(SID) :- segler(SID,_,R,_), segler(_,"Horatio",RH,_), R<=RH.
.decl nochBesser(sid: number)
nochBesser(SID) :- segler(SID,_,_,_), !dochSchlecht(SID).
.output nochBesser
```

8. Geben Sie den Namen und Alter des ältesten Seglers aus.

```
.decl junger(sid: number)
junger(SID) :- segler(SID,_,_,A), segler(_,_,_,AO), A<AO.
.decl alter(name: symbol, alter: number)
alter(S,A) :- segler(SID,S,_,A), !junger(SID).
.output alter
```

Hausaufgabe 6

Gegeben sei die nachfolgende *KindEltern*-Ausprägung für den Stammbaum-Ausschnitt der griechischen Götter und Helden:

KindEltern		
Vater	Mutter	Kind
Zeus	Leto	Apollon
Zeus	Leto	Artemis
Kronos	Rheia	Hades
Zeus	Maia	Hermes
Koios	Phoebe	Leto
Atlas	Pleione	Maia
Kronos	Rheia	Poseidon
Kronos	Rheia	Zeus
Poseidon	Amphitrite	Triton

```
.decl kindEltern(vater: symbol, mutter: symbol, kind: symbol)
kindEltern("zeus","leto","apollon").
kindEltern("zeus","leto","artemis").
kindEltern("zeus","maia","hermes").
kindEltern("koios","phoebe","leto").
kindEltern("atlas","pleione","maia").
kindEltern("kronos","rheia","hades").
kindEltern("kronos","rheia","poseidon").
kindEltern("kronos","rheia","zeus").
kindEltern("poseidon", "amphitrite", "triton").
```

Formulieren Sie folgende Anfragen in Datalog und testen Sie unter (`http://souffle.db.in.tum.de/`):

a) Bestimmen Sie alle Geschwisterpaare.

```
.decl parent(elter: symbol, kind: symbol)
parent(p,k) :- kindEltern(p,_,k).
parent(p,k) :- kindEltern(_,p,k).

.decl sibling(s1: symbol, s2: symbol)
sibling(a,b) :- parent(p,a),parent(p,b),a!=b.
.output sibling
```

b) Ermitteln Sie Paare von Cousins und Cousinen beliebigen Grades. Die Definition finden Sie auf Wikipedia.

```
.decl cousin(c1: symbol, c2: symbol)
cousin(a,b) :- parent(pa,a), parent(pb,b), sibling(pa,pb).
cousin(a,b) :- parent(pa,a), parent(pb,b), cousin(pa,pb).
.output cousin
```

c) Geben Sie alle Verwandtschaftspaare an. Überlegen Sie sich eine geeignete Definition von Verwandtschaft und setzen Sie diese in Datalog um.

```
.decl related(r1: symbol, r2: symbol)
related(a,b) :- sibling(a,b);parent(a,b);parent(b,a).
related(a,b) :- related(a,c),parent(c,b).
related(a,b) :- related(c,b),parent(c,a).
.output related
```

d) Bestimmen Sie alle Nachfahren von Kronos. Formulieren Sie die Anfrage auch in SQL, so dass sie unter HyPer ausführbar ist (online testen unter: `http://hyper-db.de/interface.html`). Sie können die Daten als Common Table Expression definieren und dann nutzen:

```
WITH RECURSIVE kindEltern(vater,mutter,kind) as (
  VALUES ('Zeus', 'Leto', 'Apollon'), ('Zeus', 'Leto', 'Artemis'),
   ('Kronos', 'Rheia', 'Hades'), ('Zeus', 'Maia', 'Hermes'),
   ('Koios', 'Phoebe', 'Leto'), ('Atlas', 'Pleione', 'Maia'),
   ('Kronos', 'Rheia', 'Poseidon'), ('Kronos', 'Rheia', 'Zeus'),
   ('Poseidon', 'Amphitrite', 'Triton')
), parent(eltern,kind) as (
  select vater, kind from kindEltern UNION select mutter, kind from kindEltern
) select * from parent where eltern='Zeus'
```

Datalog

```
.decl nachfahr(p: symbol, n: symbol)
nachfahr(p,n) :- parent(p,n).
nachfahr(p,n) :- nachfahr(p,x),nachfahr(x,n).
```

Alternativ

```
.decl nachfahr(p: symbol, n: symbol)
nachfahr(p,n) :- parent(p,n).
nachfahr(p,n) :- nachfahr(p,x),parent(x,n).
```

Anfrage für die Nachfahren von Kronos

```
.decl nachfahrVonKronos(n: symbol)
nachfahrVonKronos(n) :- nachfahr(p, n), p = "kronos".
.output nachfahrVonKronos
```

SQL

```
WITH RECURSIVE kindEltern(vater,mutter,kind) as (
VALUES ('Zeus', 'Leto', 'Apollon'), ('Zeus', 'Leto', 'Artemis'),
('Kronos', 'Rheia', 'Hades'), ('Zeus', 'Maia', 'Hermes'),
('Koios', 'Phoebe', 'Leto'), ('Atlas', 'Pleione', 'Maia'),
('Kronos', 'Rheia', 'Poseidon'), ('Kronos', 'Rheia', 'Zeus'),
('Poseidon', 'Amphitrite', 'Triton')
), parent(eltern,kind) as (
select vater, kind from kindEltern UNION select mutter, kind from kindEltern
), nachfahren(person, nachfahre) AS (
SELECT * from parent UNION ALL
SELECT n.person, p.kind FROM nachfahren n, parent p
WHERE p.eltern = n.nachfahre
)
select * from nachfahren WHERE person='Kronos'
```

Hausaufgabe (wird nicht in der Übung besprochen)

Sie fangen die folgende, mit RSA verschlüsselte Nachricht ab: 13. Sie kennen den öffentlichen Schlüssel (3,15). Wie lautet die Nachricht im Klartext? Geben Sie die komplette Herleitung an.
Alles laut Wikipedia `https://de.wikipedia.org/wiki/RSA-Kryptosystem`:

- Öffentlicher Schlüssel (e,N)
- Privater Schlüssel: (d,N)

$$N = p * q$$

mit p und q sehr großen Primzahlen. e, der sog. Verschlüsselungsexponent wird als teilerfremde Zahl zu $\phi(N)$ gewählt, wobei gilt $1 < e < \phi(N)$. $\phi(N)$ ist hierbei definiert als $\phi(N) = (p-1) * (q-1)$. d, der sog. Entschlüsselungsexponent, ist gerade das multiplikative Inverse von e bezüglich des Moduls $\phi(N)$. Die Berechnung erfolgt mittels erweiterten euklidschen Algorithmus.

Die Entschlüsselung einer verschlüsselten Nachricht C zu ihrem Klartext K erfolgt mittels der Formel

$$K = C^d \mod N.$$

Aus der Angabe wissen wir:

- $N = 15$
- $e = 3$
- $C = 13$

Wir müssen also zunächst d berechnen. Dies wäre einfach, wenn wir $\phi(N)$ wüssten. Hierzu ist die Primfaktorzerlegung von N nötig. Dies ist für die Zahl 15 äußerst einfach, es gilt $N = 5 * 3$. Damit ist $\phi(N) = 4 * 2 = 8$. Die Lösung der Kongruenz

$$e * d \equiv 1 \ mod \phi(N)$$

bzw. im konkreten Fall

$$3 * d \equiv 1 \ mod 8$$

können wir raten, indem wir alle im Bezug auf 8 teilerfremden Zahlen z betrachten, für die gilt: $1 < z < 8$.
Für $z = 3$ gilt $3 * 3 \equiv 1 \ mod 8$, womit $d = 3$ ist.
Wir entschlüsseln nun die Nachricht:

$$K = 13^3 \mod 15 = 7.$$

Der Klartext K ist also 7.

Hausaufgabe (wird nicht in der Übung besprochen)

Gegeben das folgende Schema der EDB[1]:

```
.decl product(maker: symbol, model: symbol, type: symbol)
.decl pc(model: symbol, speed: float, ram: number, hd: number, price: number)
.decl laptop(model: symbol, speed: float, ram: number,
             hd: number, screen: symbol, price: number)
.decl printer(model: symbol, color: symbol, type: symbol, price: number)
```

Beantworten Sie in Datalog und testen Sie unter (`http://souffle.db.in.tum.de/`):

a) What PC models have a speed of at least 3.00 GHz?

```
.decl fast_pc(model: symbol, speed: float, price: number)
fast_pc(x,y,p) :- pc(x,y,_,_,p), y >= 3.0.
```

b) Which manufacturers make laptops with a hard disk (hd) of at least 100 GB?

```
.decl manufacturers_100GB(maker: symbol)
manufacturers_100GB(m) :- laptop(p,_,_,d,_,_), d >= 100, product(m,p,"laptop").
```

[1] Inspiriert von `http://people.inf.elte.hu/sila/DB1English/exercise06_products.pdf`.

c) Find the model number and price of products (of any type) made by manufacturer B.

```
.decl b_prod(model: symbol, price: number)
b_prod(m,p) :- product(_, m, "pc"), pc(m,_,_,_,p).
b_prod(m,p) :- product(_, m, "laptop"), laptop(m,_,_,_,_,p).
b_prod(m,p) :- product(_, m, "printer"), printer(m,_,_,p).
```

d) Find the model numbers of all color laser printers.

```
.decl color_laser_printers(model: symbol)
color_laser_printers(m) :-printer(m,"color","laser",_).
```

e) Find those manufacturers that sell Laptops, but not PC's.

```
.decl laptop_manuf(maker: symbol)
laptop_manuf(m) :- product(m,_,"laptop"), !product(m,_,"pc").
```

f) Find those hard-disk sizes that occur in two or more PC's.

```
.decl pop_sizes(hd: number)
pop_sizes(d) :- pc(m1,_,_,d,_), pc(m2,_,_,d,_), m1!=m2.
```

g) Find those pairs of PC models that have both the same cpu speed and RAM. A pair should be listed only once, e.g., list (i,j) but not (j,i).

```
.decl sim_pc(m1: symbol, m2: symbol)
sim_pc(m1,m2) :- pc(m1,s,r,_,_), pc(m2,s,r,_,_), m1<m2.
```